

## Quelques astuces avant de commencer

### Appeler des sous-programmes : Prog


► Cette partie est assez chargée, mais les notions abordées vous sembleront plutôt évidentes. Lisez sérieusement ce passage au moins une fois.

La fonction Prog permet d'appeler un sous-programme : Prog <"nom\_programme\_basic">. Vous la trouverez dans (PRGM) + [F2] (CTL) + [F1]. On ne peut pas utiliser les chaînes de caractères pour définir le nom du sous-programme à appeler (de nombreux tests ont été faits par la communauté, sans succès).

```
Prog "SOUSPROG" //On appelle le sous-programme « SOUSPROG »
```

Lorsqu'on appelle un sous-programme, la calculatrice interprète le code depuis le début de ce dernier. Lorsqu'elle arrive à la fin du sous-programme, elle revient au programme principal à l'endroit de l'appel.

Un sous-programme peut appeler un autre sous-programme et ainsi de suite jusqu'à 10 niveaux de sous-programmes.



Source : Manuel d'utilisateur Casio Graph 100+ partie 1

Un même sous-programme peut être appelé plusieurs fois dans le programme principal, sans aucune restriction sur le nombre d'appels.

### Fonctions de contrôle : Return et Stop

Pour revenir au programme du niveau supérieur, il faut utiliser la fonction Return (onglet : « Rtrn »). Tout le code qui suit le Return dans le sous-programme est ignoré. La calculatrice continue l'exécution du programme principal à l'endroit de l'appel du sous-programme.

La fonction Stop interrompt purement l'exécution du programme. Si la fonction Stop est utilisée dans un sous-programme, il n'y a pas de retour au programme principal.

## Le jeu du Pendu

Voici le rôle des différentes chaînes de caractères (Str) utilisées dans ce Pendu :

- Str 1 : le mot mystère
- Str 2 : ce que voit la personne qui doit trouver le mot mystère
- Str 3 : liste « lettres utilisées et non valides »
- Str 6 : chaîne tampon (temporaire, utilisée pour des calculs sur les chaînes de caractères)
- Str 7 : chaîne tampon 2
- Str 8 : lettre dont on doit déterminer le nombre d'occurrences dans le mot (cf "PENDU3")
- Str 9 : les 26 lettres de l'alphabet

## Les sous-programmes

### Programme « PENDU2 »

Ce sous-programme n'affiche que l'entête du jeu.

```
ClrText
"----- PENDU -----"
" "           //un espace entre les guillemets
```

### Programme « PENDU3 »

Par contre, ce sous-programme est nettement plus intéressant pour nous. Dans le précédent cours, nous avons soulevé un problème important : StrSrc() ne donne que la position de la première occurrence. Or, un mot peut contenir plusieurs fois la même lettre. Nous allons donc améliorer cette fonction.


```
0→A
ClrList 1           //Initialisation de la liste contenant la position des occurrences
For S→B To L       //On sait que la première occurrence commence à S et L correspond à la
longueur de la chaîne (voir le programme principal « PENDU »)
If Not StrCmp(StrMid(Str 1,B,1),Str 8) //On vérifie s'il y a une occurrence
Then Isz A         //On incrémente A de 1 car List 1[0] n'existe pas
B→List 1[A]       //On stocke la position de cette occurrence dans la List 1
IfEnd
Next
For 1→B To A      //On remplace toutes les caractères « - » par la lettre contenue dans Str 8
List 1[B]
StrLeft(Str 2,Ans-1)+Str 8+StrRight(Str 2,L-Ans)→Str 2
Next
```

## Le programme principal « PENDU »

```

"ABCDEFGHIJKLMNPOQRSTUVWXYZ"→Str 9 //Initialisation de l'alphabet
Prog "PENDU2"
/* SELECTION DU MODE DU JEU */
" 1.Mode 1 joueur"
" 2.Mode 2 joueurs"
Do
Getkey→B
LpWhile B≠72 And B≠62
If B=72
Then 1→M
Else 1→J :2→M
IfEnd
//Si M=1 alors Mode= « 1 joueur », si M=2 alors Mode = « 2 joueurs »
Do //1
If M=1
Then RanInt(1,10→N
/* CHOIX D'UN MOT DANS LA BANQUE DE MOTS PAR LA CALCULATRICE */
N=1=>"CALCULATRICE"→Str 1
N=2=>"PROGRAMMATION"→Str 1
N=3=>"VACANCES"→Str 1
N=4=>"SOLEIL"→Str 1
N=5=>"RESTAURANT"→Str 1
N=6=>"RANDONNEE"→Str 1
N=7=>"MECANIQUE"→Str 1
N=8=>"SPORT"→Str 1
N=9=>"QUARTIER"→Str 1
N=10=>"COMMUNAUTE"→Str 1
Else Do //2
Prog "PENDU2"
"JOUEUR (2 espaces), VOTRE MOT ?"
Locate 8,4,1+(2-J
?→Str 1
LpWhile StrLen(Str 1)>21 //Fin 2. Le mot ne doit pas dépasser les 21 caractères
IfEnd
ClrText
""→Str 3 //Initialisation de la liste « lettres utilisées et non valides »
/* DEBUT ALGORITHME POUR CACHER LE MOT MYSTERE */
StrLen(Str 1)→L
11-Int(.5L→W
StrLeft(Str 1,1)→Str 6
Str 6→Str 2
For 1→B To L-2
Str 2+"-"→Str 2

```



```

Next
StrRight(Str 1,1)→Str 7
Str 2+Str 7→Str 2
1→S
Str 6→Str 8
Prog "PENDU3"
Str 7→Str 8
Prog "PENDU3"
/*FIN DE L'ALGORITHME */
0→Z //Initialisation du compteur de « coups manqués »
While StrCmp(Str 1,Str 2) And Z<9 //Tant que le mot mystère n'est pas trouver ou bien que
le bonhomme n'est pas pendu
Locate W,2,Str 2 //On affiche le mot mystère version « cachée »
Do //2
Do //3
.1Getkey // raccourci de « 0.1 * Getkey » ou « Getkey / 10 »
LpWhile Not Ans //Fin 3
6-10Frac Ans→R //Pour obtenir la correspondance entre le Getkey et sa valeur
8-Int Ans+5R+R(R<3→G //alphabétique
LpWhile G<1 Or G>26 //Fin 2. On vérifie que la touche appuyée est correcte
StrMid(Str 9,G,1)→Str 8 //On récupère la lettre correspondante
If Not StrSrc(Str 3,Str 8) //On vérifie si le joueur n'a pas déjà utilisée cette lettre
Then StrSrc(Str 1,Str 8)→S //On vérifie que cette lettre est présente dans le mot mystère
If S
Then Prog "PENDU3" //On vérifie le nombre d'occurrences
Else Str 3+Str 8→Str 3 //Sinon, on ajoute cette lettre à la liste « lettres utilisées »
Locate 12,6,Str 3 //On affiche cette liste
Isz Z
/* AFFICHAGE D'UNE PARTIE DU PENDU EN FONCTION DU NOMBRE DE COUPS MANQUES */
If Z=1
Then Locate 1,4,"|"
Locate 1,5,"|"
Locate 1,6,"|"
Locate 1,7,"|"
IfEnd
Z=2=>Locate 2,4,"---"
Z=3=>Locate 2,5,"/"
Z=4=>Locate 4,5,"O"
Z=5=>Locate 4,6,"|"
Z=2=>Locate 3,7,"/"
Z=2=>Locate 5,7,"\" //On doit mettre deux antislashes « \ \ » car « \ " » affiche « " ».
Z=2=>Locate 3,5,"\" //Cela permet de différencier les guillemets « " » qui délimitent
Z=2=>Locate 5,5,"/" // une chaîne du guillemet qui correspond au caractère
IfEnd
/* FIN DE L'AFFICHAGE*/

```

```

IfEnd
WhileEnd //Fin du « Tant que »
ClrText
If Z=9
Then "VOUS N'AVEZ PAS TROUVE LE MOT MYSTERE..."
Else "VOUS AVEZ TROUVE LE MOT MYSTERE !"
IfEnd
Locate W,5,Str 1[DISP]
Prog "PENDU2"
" 1.Continuer"
" 2.Quitter"
Do
Getkey
LpWhile Ans≠72 And Ans≠62
If Ans=72 And M=2
Then Isz J
J=3=>1→J
IfEnd
LpWhile Ans=72 //Fin 1. La partie continue tant que le(s) joueur(s) le décide(nt)

```

```

----- PENDU -----
1.Mode 1 joueur
2.Mode 2 joueurs

```

```

P-----N

```

```

PRO-RA--A-ION
|---|
| / 0 |
| |   |
| \   |
|---|
EUYVFL

```

```

VOUS AVEZ TROUVE LE
MOT MYSTERE !

PROGRAMMATION

```

## Ce cours de programmation Basic Casio est fini !

A travers ce cours de programmation, nous voulions vous montrer les capacités et les limites du langage Basic Casio et des calculatrices graphiques. Il y a beaucoup de fonctions, notamment tout ce qui concerne les graphismes, dont je n'en ai pas parlé. En effet, les graphismes ne sont pas le point fort de langage, surtout avec les dernières Fx-CG 10/20 où il faut attendre une bonne demi-seconde pour tracer une ligne (10 fois plus vite sur les Graph 75/95 SD !!).

Le Basic Casio n'est pas un langage standardisé et l'interpréteur Basic comporte des différences plus ou moins importantes selon les modèles. Certaines astuces fonctionnent uniquement sur certaines calculatrices. Pour autant, malgré ces limites, de nombreux programmeurs développent des jeux de toute beauté, avec par exemples, des animations (combats, mouvements etc...).

La communauté francophone des programmeurs Casio est très majoritairement regroupée sur Planète-Casio : <http://www.planete-casio.fr>. Le site existe depuis 8 ans et contient une énorme base de programmes, plus de 1700 au moment où nous terminons ce cours, ainsi que de nombreux tutoriels et astuces. On organise régulièrement des concours de programmation en partenariat avec Casio Education : <http://www.casio-education.fr>.

La programmation sur calculatrice ne se limite pas au Basic Casio, il y a aussi le C/C++ et le LuaFX.

Voici d'autres communautés. Certaines sont mixtes, c'est-à-dire, qu'on parle aussi d'autres marques :

- Omnimaga : <http://www.omnimaga.org> (mixte, anglais, français, allemand)
- Cemetech : <http://www.cemetech.net> (mixte, anglais)
- Universal Casio Forum (UCF): <http://community.casiocalc.org> (Casio, anglais, portugais)
- JeuxCasio : <http://www.jeuxcasio.com> (Casio, français)
- et une multitude de sites plus petits, à caractère personnel...

A bientôt sur [Casio Calculatrices](#) et [Planète-Casio](#) !